

TopSpeed ODBC Interface — syntax

statement ::= CREATE create | DROP drop | SELECT select orderby | INSERT insert | DELETE delete | UPDATE update | passthroughSQL

passthroughSQL ::= any statement supported by the backend

create ::= TABLE tablename (createcols) | INDEX indexname ON tablename (indexcolumns)

indexcolumns ::= indexcolumn | indexcolumn , indexcolumns

indexcolumn ::= columnname asc

drop ::= TABLE tablename | INDEX indexname

select ::= selectcols FROM tablelist where groupby having

delete ::= FROM tablename where

insert ::= INTO tablename insertvals

update ::= tablename SET setlist where

setlist ::= set | setlist , set

set ::= columnname = NULL | columnname = expression

insertvals ::= (columnlist) VALUES (valuelist) | VALUES (valuelist) | (columnlist) VALUES (SELECT select) | VALUES (SELECT select)

columnlist ::= columnname , columnlist | columnname

valuelist ::= NULL , valuelist | expression , valuelist | expression | NULL

selectcols ::= selectallcols * | selectallcols selectlist

selectallcols ::= | ALL | DISTINCT

where ::= | WHERE boolean

having ::= | HAVING boolean

boolean ::= and | and OR boolean

and ::= not | not AND and

not ::= comparison | NOT comparison

comparison ::= (boolean) | colref IS NULL | colref IS NOT NULL | expression LIKE pattern | expression NOT LIKE pattern | expression IN (valuelist) | expression NOT IN (valuelist) | expression op expression | EXISTS (SELECT select) | expression op selectop (SELECT select) | expression IN (SELECT select) | expression NOT IN (SELECT select) | expression BETWEEN expression AND expression | expression NOT BETWEEN expression AND expression



TopSpeed ODBC Interface — syntax

pattern ::= string | ? | USER

expression ::= expression + times | expression - times | times

times ::= times * neg | times / neg | neg

neg ::= term | + term | - term

term ::= (expression) | colref | simpleterm | aggterm | scalar

scalar ::= scalarescape | scalarshorthand

scalarescape ::= —*(VENDOR(MICROSOFT),PRODUCT(ODBC) FN fn)*—

scalarshorthand ::= { FN fn }

fn ::= functionname (valuelist) | functionname ()

aggterm ::= COUNT (*) | AVG (expression) | MAX (expression) | MIN (expression) | SUM (expression) | COUNT (expression)

simpleterm ::= string | realnumber | ? | USER | date | time | timestamp

groupby ::= | GROUP BY groupbyterms

groupbyterms ::= colref | colref , groupbyterms

orderby ::= | ORDER BY orderbyterms

orderbyterms ::= orderbyterm | orderbyterm , orderbyterms

orderbyterm ::= colref asc | integer asc

asc ::= | ASC | DESC

colref ::= aliasname . columnname | columnname

tablelist ::= tablelistitem , tablelist | tablelistitem

tablelistitem ::= tableref | outerjoin

outerjoin ::= ojescape | ojshorthand

ojescape ::= —*(VENDOR(MICROSOFT),PRODUCT(ODBC) OJ oj)*—

ojshorthand ::= { OJ oj }

oj ::= tableref LEFT OUTER JOIN tableref ON boolean | tableref LEFT OUTER JOIN
oj ON boolean

